
Heltalsdivision genom upprepad subtraktion Y(16)/X(8) = Q(16) + R(8)/X(8) 
Dividenden Y(16) finns från början i M(YHI):M(YLO) (Hög del på adress YHI och låg del på adress YLO) 
Divisorn X(8) finns i M(Xval).  (X ≠ 0) 

Kvoten Q(16) finns efter divisionen i M(QHI):M(QLO)   (Hög del på adress QHI och låg del på adress QLO) 
Resten R(8) finns efter divisionen i M(REST) 

LOOP  LDA  YLO  Hämta låg byte av dividend 
   SUBA  Xval  Subtrahera divisor 
   STA  YLO  Uppdatera YLO 

   LDA  YHI  Hämta hög byte av dividend 
   SBCA  #0   Drag bort ev borrow från YHI 
   STA  YHI  Uppdatera YHI 

   BLO  READY Division färdig 

   INC  QL O  Ej klart, öka låg byte av kvoten med 1 
   BNE  LOOP  Fortsätt om ej carry till hög byte av kvoten 

   INC  QHI  Carry till hög byte 
   BRA  LOOP 

READY LDA  YL   Hämta rest (nu negativ) 
   ADDA Xval  Justera rest (addera divisor) 
   STA  REST  Uppdatera rest 
     …      Färdigt 

 

Heltalsdivision genom subtraktion och skift Y(16)/X(8) = Q(16) + R(8)/X(8) 
Dividenden Y(16) finns från början i M(YHI):M(YLO) (Hög del på adress YHI och låg del på adress YLO) 
Divisorn X(8) finns i M(Xval).  (X ≠ 0) 

Kvoten Q(16) finns efter divisionen i M(QHI):M(QLO)   (Hög del på adress QHI och låg del på adress QLO) 

Resten R(8) finns efter divisionen i register A och M(REST) 

Register A har från början värdet 0 
M(COUNT) är en skifträknare, som från början har värdet 16 

LOOP  LSL  YLO  2*dividend (msb till C-flaggan) 
   ROL  YHI         -" - 

   ROLA     MS-bit till A (C-flaggan till A-registret) 
   BCS  SUBTR Overflow i A-reg! Subtrahera divisor från dividendens ms-bitar 

   CMPA Xval  Skall vi subtrahera divisorn från ms-bitar från dividenden? 
   BHS  SUBTR Ja! 

   LSL  QLO  Nej, skifta kvoten vänster med nolla in från höger 
   ROL  QHI  Hög byte 
   BRA  NEXT 

SUBTR SUBA  Xval  Utför subtraktionen. Blir korrekt även vid overflow! 
   ORCC  #1   Sätt carry (ger etta i kvoten) 
   ROL  QLO  Skifta kvoten vänster med etta in från höger 
   ROL  QHI              -" - 

NEXT  DEC  COUNT Skifträknare 
   BNE  LOOP  Färdigt? Nej 

   STA  REST  Ja. Uppdatera rest 
     …      Färdigt 


